

limit (and Limit)

The `limit` command is used to compute limits (what else?). The syntax of `limit` is Maple's usual

```
limit(what, how );
```

syntax. "What" in this case refers to "take the limit of what?", and "how" to "as what variable approaches what value?". For example, to compute

$$\lim_{x \rightarrow 2} \frac{3x - 6}{x^2 - 4}$$

we would enter:

```
> limit((3*x-6)/(x^2-4),x=2);
```

$$\frac{3}{4}$$

Notice, the first argument of `limit` is the expression whose limit is being taken, and the second tells which variable approaches what number. If the limit does not exist, as in

```
> limit(1/x,x=0);
```

undefined

Maple tells you.

Occasionally Maple will be unable to determine a limit (or whether it exists). In such a case, Maple will return nothing -- when this happens, you can give Maple an assist sometimes by setting a variable called "Order" equal to a whole number somewhat bigger than 6 (its default variable) -- this is because Maple uses series to compute limits (you will see or have seen how to do this in Math 141/151), and Maple usually uses only the first six terms of the series. When you change Order, you enable Maple to do more accurate (although more time-consuming)

calculations. It is somewhat like computing to more decimal places.

The limits Maple takes are "two-sided, real limits". This means Maple assumes that when you type $x=15$ as the second argument of `limit`, you mean x should approach 15 from either above or below through real values only (as opposed to complex ones). Maple can compute one-sided and complex limits. For example

```
> limit(1/x,x=0,right);
```

∞

You could also do limits from the left.

It is possible to compute limits as x tends to infinity (or -infinity):

```
> limit(arctan(x),x=infinity);
```

$\frac{1}{2}\pi$

Finally, limits of functions of more than one variable are allowed (for Math 141, 151 and 240):

```
> limit(x/(x^2+y^2),{x=0,y=0});
```

undefined

Remarks: It is always a good idea to use `plot` in conjunction with `limit` -- you can often get an intuition for the limit from the plot.

It is instructive to use `limit` for computing derivatives and integrals from the definition.

Occasionally, to make your worksheets easier to read, you may wish to have Maple display a limit in standard mathematical notation without evaluating it. For this there is a capitalized, "inert" form of the limit command:

```
> Limit(exp(x)/(1-x),x=0);
```

$$\lim_{x \rightarrow 0} \frac{e^x}{1-x}$$

Sometimes, you can use the two forms together to produce meaningful sentences:

```
> Limit(exp(x)/(1-x),x=0)=limit(exp(x)/(1-x),x=0);
```

$$\lim_{x \rightarrow 0} \frac{e^x}{1-x} = 1$$

Few things can go wrong using the `limit` command, other than syntax errors -- except possibly that sometimes the variable in the `limit` command (the `x` above) has already been given a value that you forgot about:

```
> x:=3;
```

```
x := 3
```

```
> limit(x^2/sin(2*x^2),x=0);
```

```
Error, (in limit) invalid arguments
```

Here, `limit` is objecting to your trying to let 3 approach 0.